Math 132
Fall 2005 Exam III

1. Calculate $\displaystyle\int_{-2}^{1} \frac{x}{\sqrt{5-4x-x^2}}\,dx$.

a) $\pi - 1$ **b)** $\pi - 2$ **c)** $\pi - 3$ **d)** $\pi - 4$ **e)** $\pi - 5$
f) $1 - \pi$ **g)** $2 - \pi$ **h)** $3 - \pi$ **i)** $4 - \pi$ **j)** $5 - \pi$

Solution: h

```
> with(student):
> J := Int(x/sqrt(5-4*x-x^2),x = -2 .. 1);
```

$$J := \int_{-2}^{1} \frac{x}{\sqrt{5 - x^2 - 4x}}\,dx$$

```
> A := numer(integrand(J));  B := denom(integrand(J));
```

$$A := x$$
$$B := \sqrt{5 - x^2 - 4x}$$

```
> C := completesquare(B,x);
```

$$C := \sqrt{9 - (x+2)^2}$$

```
> eqn := x + 2 = 3*sin(t);
```

$$eqn := x + 2 = 3\sin(t)$$

```
> J1 := changevar(eqn, J, t);
```

$$J1 := \int_{0}^{\frac{\pi}{2}} \frac{3\,(-2 + 3\sin(t))\cos(t)}{\sqrt{13 - (-2 + 3\sin(t))^2 - 12\sin(t)}}\,dt$$

```
> J2 := map(simplify, J1 );
```

$$J2 := \int_{0}^{\frac{\pi}{2}} (-2 + 3\sin(t))\,\mathrm{csgn}(\cos(t))\,dt$$

```
> plot([Re(csgn(cos(t))),Im(csgn(cos(t)))], t= 0..2*Pi, color =
  [RED, BLUE], discont = true);

  #In simplifying the square root, Maple has introduced the
  "complex sign" of cos(t). In fact, over the interval of
  integration, this factor is 1. This plot is intended to show
  that. Stripped of this term, the integrand is elementary.
```




```
> value(J2);
```

$$-\pi + 3$$

2. The triangular region in the first quadrant bounded by $y = 2 - x^2$, $y = x$, and $x = 0$ is rotated about the line y = 3. What is the volume of the solid of revolution that results?

a) $\frac{10\pi}{3}$ **b)** $\frac{14\pi}{3}$ **c)** $\frac{9\pi}{2}$ **d)** $\frac{15\pi}{2}$ **e)** 4π

f) $\frac{61\pi}{15}$ **g)** $\frac{67\pi}{15}$ **h)** $\frac{24\pi}{5}$ **i)** $\frac{28\pi}{5}$ **j)** 6π

Solution: g

```
> plot([x,2-x^2,3,6-x, 4+x^2], x = 0 .. 1, color =
  [BLUE,BLUE,BLACK,COLOR(RGB,.75,.75,.75),COLOR(RGB,.75,.75,.75)]
  , view=[0..1,0..6]);
```




```
> volume := Pi*int((3-x)^2-(3-(2-x^2))^2, x = 0 .. 1);
```

$$volume := \frac{67\,\pi}{15}$$

3. Calculate the arc length of $y = \frac{x^3}{3} + \frac{1}{4\,x}$ **for** $1 \le x \le 2$.

a) $\frac{58}{23}$ b) $\frac{59}{24}$ c) $\frac{12}{5}$ d) $\frac{61}{26}$ e) $\frac{62}{27}$

f) $\frac{63}{28}$ g) $\frac{64}{29}$ h) $\frac{11}{6}$ i) $\frac{66}{31}$ j) $\frac{67}{32}$

Solution: b

```
> f := x -> x^3/3 + 1/4/x;
```

$$f := x \rightarrow \frac{1}{3}x^3 + \frac{1}{4}\frac{1}{x}$$

```
> toBeIntegrated := simplify(sqrt(1+diff(f(x),x)^2))  assuming x
  > 0;
```

$$toBeIntegrated := \frac{1+4x^4}{4x^2}$$

```
> int(toBeIntegrated, x=1..2);
```

$$\frac{59}{24}$$

4. The graph of $y = \frac{x}{3}$ for $3 \le x \le 6$ is rotated about the x-axis. What is the surface area of the resulting figure?

a) $2\pi\sqrt{3}$ b) $4\pi\sqrt{3}$ c) $5\pi\sqrt{3}$ d) $6\pi\sqrt{3}$ e) $2\pi\sqrt{5}$
f) $3\pi\sqrt{5}$ g) $4\pi\sqrt{5}$ h) $2\pi\sqrt{10}$ i) $3\pi\sqrt{10}$ j) $6\pi\sqrt{2}$

Solution: i

```
> f := x ->  x/3;
```

$$f := x \rightarrow \frac{1}{3}x$$

```
> answer := 2*Pi*int(f(x)*sqrt(1+diff(f(x),x)^2),x=3..6);
```

$$answer := 3\pi\sqrt{10}$$

5. Let R be the region bounded by the graph of $y = 4 - x^2$, the x-axis, and the vertical line $x = -1$. What is the x-coordinate of the center of mass of R?

a) $\frac{1}{4}$ b) $\frac{3}{4}$ c) $\frac{5}{4}$ d) $\frac{1}{5}$ e) $\frac{1}{3}$

f) $\frac{2}{5}$ g) $\frac{3}{8}$ h) $\frac{3}{5}$ i) $\frac{5}{8}$ j) $\frac{8}{15}$

Solution: a

```
> f := x -> 4-x^2;
```

$$f := x \to 4 - x^2$$

```
> x_bar := int(x*f(x),x=-1..2)/int(f(x),x=-1..2);
```

$$x\_bar := \frac{1}{4}$$

6. What is the y-coordinate of the center of mass of the region R of the preceding question?

a) 1 b) $\frac{3}{2}$ c) $\frac{17}{10}$ d) $\frac{9}{5}$ e) 2

f) $\frac{5}{2}$ g) $\frac{5}{3}$ h) $\frac{8}{3}$ i) $\frac{8}{5}$ j) $\frac{15}{8}$

Solution: c

```
> int(f(x)^2,x=-1..2)/2/int(f(x),x=-1..2);
```

$$\frac{17}{10}$$

7. Suppose that $f(x) = x^2$. On the interval [1 , b], f(x) assumes its average value when $x = \sqrt{19}$.

What is b ?

a) 5 b) $3\sqrt{3}$ c) $4\sqrt{2}$ d) 6 e) $5\sqrt{2}$

f) $4\sqrt{3}$ g) $2\sqrt{10}$ h) $3\sqrt{5}$ i) 7 j) $2\sqrt{14}$

Solution: i

```
> a :=1;   f := x -> x^2;
```

$$a := 1$$

$$f := x \to x^2$$

```
> f_ave := int(f(x),x = a .. b)/(b-a);
```

$$f\_ave := \frac{\frac{b^3}{3} - \frac{1}{3}}{b - 1}$$

```
> simplify(f_ave);   #The average is actually quadratic in  b
```

$$\frac{b^2}{3} + \frac{b}{3} + \frac{1}{3}$$

```
> b = solve(f(sqrt(19)) = f_ave, b);
```

$$b = (-8, 7)$$

8. For what constant c is $f(x) = c\,x^2$ a probability density function for $1 \le x \le 2$.

a) $\frac{1}{4}$ b) $\frac{1}{2}$ c) $\frac{2}{3}$ d) 1 e) $\frac{2}{9}$

f) $\frac{4}{3}$ g) $\frac{1}{3}$ h) $\frac{3}{4}$ i) $\frac{3}{7}$ j) $\frac{3}{8}$

Solution: i

```
> eqn := int(c*x**2, x=1..2) = 1;
```

$$eqn := \frac{7\,c}{3} = 1$$

```
> c = solve(eqn, c);
```

$$c = \frac{3}{7}$$

9. A random variable assumes values in the interval $[0, 2]$ and has probability density function $f(x) = \frac{x^3}{4}$ for $0 \le x \le 2$. What is the mean of X?

a) $\frac{1}{2}$ b) $\frac{2}{3}$ c) $\frac{3}{4}$ d) $\frac{5}{8}$ e) $\frac{8}{5}$

f) $\frac{15}{16}$ g) $\frac{16}{15}$ h) $\frac{8}{7}$ i) $\frac{3}{2}$ j) $\frac{4}{3}$

Solution: e

```
> mu = int(x*(x^3/4),x = 0 .. 2);
```

$$\mu = \frac{8}{5}$$

10. A random variable X assumes values in the interval $\left[0, \frac{\pi}{2}\right]$ and has probability density function $f(x) = \cos(x)$ for $0 \le x \le \frac{\pi}{2}$. What is the value of $P\left(\frac{\pi}{6} \le X \le \frac{\pi}{3}\right)$?

a) $\frac{\sqrt{2}-1}{2}$ b) $\frac{\sqrt{2}-1}{4}$ c) $\frac{\sqrt{3}-1}{2}$ d) $\frac{\sqrt{3}-1}{4}$ e) $\frac{2-\sqrt{2}}{2}$

f) $\frac{2\sqrt{2}-1}{2}$ g) $\frac{2\sqrt{2}-1}{4}$ h) $\frac{\sqrt{3}-\sqrt{2}}{2}$ i) $\frac{2\sqrt{3}-1}{4}$ j) $\frac{\sqrt{3}-\sqrt{2}}{4}$

Solution: c

```
> int(cos(x),x = Pi/6 .. Pi/3);
```

$$\frac{\sqrt{3}}{2} - \frac{1}{2}$$

11. If m is the median of a random variable with pdf given by $f(x) = \frac{x^3}{4}$ for $0 \le x \le 2$,

then what is $\log_2(m)$?

a) 1/8 b) 1/4 c) 3/8 d) 1/2 e) 5/8
f) 3/4 g) 7/8 h) 5/16 i) 7/16 j) 9/16

Solution: f

```
> eqn := int(x^3/4, x=0..m) = 1/2;
```

$$eqn := \frac{m^4}{16} = \frac{1}{2}$$

```
> m = solve(eqn, m);
```

$$m = (2^{(3/4)}, 2^{(3/4)} I, -2^{(3/4)}, -I\, 2^{(3/4)})$$

12. In stretching a spring 60 cm beyond equilibrium, 1.44 J work is done. How many newtons of force are needed to maintain the spring at that position?

a) 3.8 b) 4.0 c) 4.2 d) 4.4 e) 4.6
f) 4.8 g) 5.0 h) 5.2 i) 5.4 j) 5.6

Solution: f

```
> eqn1 := W = int(k*x, x = 0 .. 6/10);
```

$$eqn1 := \mathrm{W} = \frac{9\,k}{50}$$

```
> eqn2 := subs( W = 1.44, eqn1);
```

$$eqn2 := 1.44 = \frac{9\,k}{50}$$

```
> eqn3 := k = solve(eqn2, k);
```

$$eqn3 := k = 8.$$

```
> HookesLaw := F = k*x;
```

$$HookesLaw := F = k\,x$$

```
> subs({eqn3, x=6/10}, HookesLaw);
```

$$F = 4.800000000$$

13. A hemispherical tank completely filled with water is pumped dry. The radius of the disc that is the top part of the tank is 2m. At a depth y meters below the top of the tank, the horizontal cross-section of the tank has area $\pi\,(4 - y^2)$. What is the work done ?

a) $1200\,\pi\,g$ b) $1600\,\pi\,g$ c) $2000\,\pi\,g$ d) $2400\,\pi\,g$ e) $2800\,\pi\,g$
f) $3200\,\pi\,g$ g) $3600\,\pi\,g$ h) $4000\,\pi\,g$ i) $4400\,\pi\,g$ j) $4800\,\pi\,g$

Solution: h

```
> A := 1000*g*int(Pi*y*(4-y^2),y = 0 .. 2);
```

$$A := 4000\,g\,\pi$$

14. A thick rope hangs over the side of a 100 foot tall building, but does not reach the ground. The rope weighs 24 pounds per foot. The dangling end of the rope is tied to a 120 pound sack

of sugar. In pulling the sack of sugar all the way to the top of the building, a hungry, motivated ant expends 4992 foot-pounds of work. How may feet of rope were hanging over the side of the building?

a) 10 b) 11 c) 12 d) 13 e) 14
f) 15 g) 16 h) 17 i) 18 j) 19

Solution: g

```
> eqn := int(24*y,y = 0 .. a)+120*a = 4992;
```

$$eqn := 12\,a^2 + 120\,a = 4992$$

```
> eqn2 := map(z -> z/12, eqn);
```

$$eqn2 := a^2 + 10\,a = 416$$

```
> solve(eqn2);
```

16, -26

15. Calculate $\int_0^{16} \frac{1}{(16-x)^{\left(\frac{1}{4}\right)}}\,dx$.

a) 21/2 b) 25/2 c) 16/5 d) 32/3 e) 32/5
f) 9/2 g) 16/3 h) 17/4 i) 21/4 j) 64/5

Solution: d

```
> simplify(int(1/((16-x)^(1/4)),x = 0 .. 16));
```

$$\frac{32}{3}$$

16. Calculate $\displaystyle\int_0^{\frac{\pi}{4}} \frac{\sec(x)^2}{\tan(x)^{\left(\frac{1}{3}\right)}}\,dx$.

a) $\frac{1}{2}$ b) $\frac{2}{3}$ c) $\frac{3}{4}$ d) $\frac{5}{8}$ e) $\frac{8}{5}$

f) $\frac{15}{16}$ g) $\frac{16}{15}$ h) $\frac{8}{7}$ i) $\frac{3}{2}$ j) $\frac{4}{3}$

Solution: i

```
> with(student):
> J := Int(sec(x)^2/(tan(x))^(1/3), x = 0 .. Pi/4);
```

$$J := \int_0^{\frac{\pi}{4}} \frac{\sec(x)^2}{\tan(x)^{(1/3)}}\,dx$$

```
> K := changevar(u=tan(x), J, u);
```

$$K := \int_0^1 \frac{1}{u^{(1/3)}}\,du$$

```
> value(K);
```

$$\frac{3}{2}$$

17. Calculate $\displaystyle\int_2^{\infty} \frac{x^2}{(x^3-7)^{\left(\frac{5}{3}\right)}}\,dx$.

a) $\frac{1}{2}$ b) $\frac{2}{3}$ c) $\frac{3}{2}$ d) $\frac{3}{4}$ e) 1

f) $\frac{4}{3}$ g) $2\sqrt{2}$ h) $\frac{\sqrt{2}}{2}$ i) $\frac{\sqrt{2}-1}{2}$ j) 2

Solution: a

```
> int(x^2/((x^3-7)^(5/3)),x = 2 .. infinity);
```

$$\frac{1}{2}$$

18. Evaluate $\int_{\frac{2}{\pi}}^{\infty} \frac{x\sin\left(\frac{2}{x}\right) - 2\cos\left(\frac{2}{x}\right)}{x}\,dx$. You may find the following formula useful:

$$\frac{d}{dx}\left(x\sin\left(\frac{2}{x}\right)\right) = \frac{x\sin\left(\frac{2}{x}\right) - 2\cos\left(\frac{2}{x}\right)}{x}.$$

a) 1 b) −1 c) 2 d) −2 e) 3

f) −3 g) 4 h) −4 i) 5 j) −5

Solution: c

```
> eqn := Int((x*sin(2/x)-2*cos(2/x))/x,x = 2/Pi .. infinity) =
  Limit(subs(x=N, x*sin(2/x) ) - subs(x=2/Pi, x*sin(2/x) ), N =
  infinity);
```

$$eqn := \int_{\frac{2}{\pi}}^{\infty} \frac{x\sin\left(\frac{2}{x}\right) - 2\cos\left(\frac{2}{x}\right)}{x}\,dx = \lim_{N\to\infty} N\sin\left(\frac{2}{N}\right) - \frac{2\sin(\pi)}{\pi}$$

```
> eqn2 := lhs(eqn) = simplify( rhs(eqn) );
```

$$eqn2 := \int_{\frac{2}{\pi}}^{\infty} \frac{x \sin\left(\frac{2}{x}\right) - 2\cos\left(\frac{2}{x}\right)}{x} \, dx = \lim_{N \to \infty} N \sin\left(\frac{2}{N}\right)$$

```
> eqn3 := lhs(eqn2) = Limit(sin(2*u)/u, u = 0); #Change variable: u=1/N
```

$$eqn3 := \int_{\frac{2}{\pi}}^{\infty} \frac{x \sin\left(\frac{2}{x}\right) - 2\cos\left(\frac{2}{x}\right)}{x} \, dx = \lim_{u \to 0} \frac{\sin(2\,u)}{u}$$

```
> eqn4 := lhs(eqn2) = Limit(Diff(sin(2*u),u)/Diff(u,u), u = 0);
  #L'Hopital
```

$$eqn4 := \int_{\frac{2}{\pi}}^{\infty} \frac{x \sin\left(\frac{2}{x}\right) - 2\cos\left(\frac{2}{x}\right)}{x} \, dx = \lim_{u \to 0} \frac{\frac{d}{du}\sin(2\,u)}{\frac{d}{du}u}$$

```
> eqn4 := lhs(eqn2) = value( rhs( eqn3) );
```

$$eqn4 := \int_{\frac{2}{\pi}}^{\infty} \frac{x \sin\left(\frac{2}{x}\right) - 2\cos\left(\frac{2}{x}\right)}{x} \, dx = 2$$

19. Calculate the fourth partial sum of $\sum_{n=1}^{\infty} \frac{1}{2\,n\,(n+1)}$.

a) 1/10 b) 1/5 c) 3/10 d) 2/5 e) 1/2

f) 3/5 g) 7/10 h) 4/5 i) 9/10 j) 1

Solution: d

```
> a := n ->  1/2/n/(n+1);
```

$$a := n \rightarrow \frac{1}{2}\frac{1}{n(n+1)}$$

```
> sum(a(n), n = 1 .. 4);
```

$$\frac{2}{5}$$

20. Calculate $\sum_{n=1}^{\infty} \frac{3^n}{5^{(n+1)}}$ **.**

a) 1/10 b) 1/5 c) 3/10 d) 2/5 e) 1/2

f) 3/5 g) 7/10 h) 4/5 i) 9/10 j) 1

Solution: c

```
> sum(3^n/(5^(n+1)),n = 1 .. infinity);
```

$$\frac{3}{10}$$